PERCENT FEDERAL LAND FOR OIL/GAS FIELD OUTLINES 

The VBA code below calculates the area percent of a first polygon layer (e.g. oil/gas field outlines) that are within a second polygon layer (e.g. federal land) and writes out the fraction as an attribute for the first polygon layer.  If you make buffered well field outline polygons using the VBA code in BUFFERED_WELL_FIELD_OUTLINES.doc, you will have a feature class with the attribute PCTFEDLAND to use as the first polygon layer. If not, add the field PCTFEDLAND (double) to your polygon layer. You can insert multiple layers of federal land into a group layer named Federal_land as your second layer. 

Copy the code into a VBA module in ArcMap.

Inputs:

1) Data frame in ArcMap named “Task 2”

2) Layer (0) is a reference layer of your choice

3) Layer (1) is the first polygon layer (e.g. buffered well oil-field boundaries) with a field “PCTFEDLAND”

4) Layer (2) is the second polygon layer, a group layer named "Federal_land" with sublayers each having federal land polygons in the display table

Output: Layer (1)'s PctFedLand column gets updated

Code by Kirk Kuykendall, AmberGIS; text by Sam Limerick, Z, Inc.

==============================================================

Option Explicit

Sub Task2()

    '------------------------------------------------------------------

    ' Inputs:

    '  A dataframe named "Task 2"

    ' Layer (0) has a reference layer of your choice

    ' Layer (1) has buffered well polygons

    ' Layer (2) a group layer named "Federal_land" with sublayers each

    '          having federal land polygons in the "display table"

    '

    ' Outputs:

    ' Layer (1)'s PctFedLand column gets updated

    '------------------------------------------------------------------

    '  This module loop through each buffered oil field and

    '  updates the "PctFedLand" column by finding the total

    '  area of federal land from all layers in the group layer

    '  Note that merging all the coverages from these layers into

    '  one shape file or geodatabase can cause problems with

    '  "non-simple" geometry, preventing ITopologicalOperator.Intersect

    '  from working

    '------------------------------------------------------------------

    Dim pMxDoc As IMxDocument

    Set pMxDoc = ThisDocument

    Dim pMap As IMap

    Set pMap = GetMap("Task 2")

    If pMap Is Nothing Then

        MsgBox "cannot find Task 2 dataframe"

        Exit Sub

    End If

    Dim pWellBuffLayer As IFeatureLayer

    Set pWellBuffLayer = pMap.Layer(1)

    If pWellBuffLayer.FeatureClass.ShapeType <> esriGeometryPolygon Then

        MsgBox pMxDoc.FocusMap.Layer(1).Name & " is not a polygon layer"

        Exit Sub

    End If

    Dim pFedGroupLayer As ICompositeLayer

    If Not TypeOf pMap.Layer(2) Is ICompositeLayer Then

        MsgBox pMap.Layer(2).Name & " is not a group layer!"

        Exit Sub

    Else

        Set pFedGroupLayer = pMap.Layer(2)

    End If

    Dim pSR As ISpatialReference

    Set pSR = pMap.SpatialReference

    Dim pInFCur As IFeatureCursor

    Set pInFCur = pWellBuffLayer.FeatureClass.Update(Nothing, False)

    Dim lPctCol As Long

    lPctCol = pInFCur.FindField("PctFedLand")

    If lPctCol = -1 Then

        MsgBox "PctFedLand not found on " & pWellBuffLayer.FeatureClass.AliasName

        Exit Sub

    End If

    Dim lNameFld As Long

    lNameFld = pWellBuffLayer.FeatureClass.FindField(pWellBuffLayer.DisplayField)

    Dim pInFeat As IFeature

    Set pInFeat = pInFCur.NextFeature

    Do Until pInFeat Is Nothing

        Dim dTotalFedLand As Double

        dTotalFedLand = 0#

        Dim l As Long

        Dim lLayersIntersected As Long

        lLayersIntersected = 0

        For l = 0 To pFedGroupLayer.Count - 1

            Application.StatusBar.Message(0) = "... intersecting " & pInFeat.Value(lNameFld) & " with " & _

                                               pFedGroupLayer.Layer(l).Name

            Dim dFedLand As Double, pFedPoly As IGeometry

            Set pFedPoly = GetFed(pInFeat.ShapeCopy, pFedGroupLayer.Layer(l), _

                              pMap.SpatialReference)

            pFedPoly.Project pMap.SpatialReference

            If Not pFedPoly.IsEmpty Then

                lLayersIntersected = lLayersIntersected + 1

                'modDraw.DrawGeometry pMxDoc.FocusMap, pFedPoly, vbRed

                dFedLand = GetArea(pFedPoly)

'              Debug.Print pFedGroupLayer.Layer(l).Name & ": " & dFedLand

                dTotalFedLand = dTotalFedLand + dFedLand

            End If

        Next l

        Dim dPctFedLand As Double, dBuffArea As Double

        dBuffArea = GetArea(pInFeat.ShapeCopy)

        dPctFedLand = dTotalFedLand / dBuffArea

        Application.StatusBar.Message(0) = pInFeat.Value(lNameFld) & "," _

                    & Format(dPctFedLand, "00.0%") _

                    & ", Acres Fed Land: " & Format(dTotalFedLand, "000") _

                    & ", Acres OilGas Field: " & Format(dBuffArea, "0000") & ", "

        pInFeat.Value(lPctCol) = dPctFedLand

        pInFCur.UpdateFeature pInFeat

        Set pInFeat = pInFCur.NextFeature

    Loop

    Application.StatusBar.Message(0) = "Task 2 finished"

End Sub

Function GetSRName(pGeom As IGeometry) As String

    GetSRName = pGeom.SpatialReference.Name

End Function

Function GetSR(pFC As IFeatureClass) As ISpatialReference

    ' get the spatial reference from the geometry column

    Set GetSR = pFC.Fields.Field(pFC.FindField(pFC.ShapeFieldName)).GeometryDef.SpatialReference

End Function

Function GetFed(pInGeom As IGeometry, pFLayer As IFeatureLayer, _

                pOutSR As ISpatialReference) As IPolygon

    ' project the search geometry (well buffer polygon) into

    ' whatever the projection of the federal land featureclass is

    pInGeom.Project GetSR(pFLayer.FeatureClass)

    Simplify pInGeom

    Dim pSF As ISpatialFilter

    Set pSF = New SpatialFilter

    Set pSF.Geometry = pInGeom

    pSF.SpatialRel = esriSpatialRelIntersects

    Dim pDispTable As IDisplayTable

    Set pDispTable = pFLayer

    Dim pFCur As IFeatureCursor

    Set pFCur = pDispTable.SearchDisplayTable(pSF, False)

    Dim pFedPoly As IGeometryCollection

    Set pFedPoly = New Polygon

    Dim pOutGeom As IGeometry

    Set pOutGeom = pFedPoly

    Set pOutGeom.SpatialReference = GetSR(pFLayer.FeatureClass)

    Dim pFeat As IFeature

    Set pFeat = pFCur.NextFeature

    Do Until pFeat Is Nothing

        Dim pPartPoly As IPolygon

        Set pPartPoly = ClipPoly(pInGeom, pFeat.ShapeCopy)

        If Not pPartPoly.IsEmpty Then

            pFedPoly.AddGeometryCollection pPartPoly

        End If

        Set pFeat = pFCur.NextFeature

    Loop

'    Dim pOutGeom As IGeometry

'    Set pOutGeom = pFedPoly

'    'pOutGeom.Project pOutSR ' project into utm

    Set GetFed = pFedPoly

End Function

Function GetArea(pArea As IArea) As Double

    Dim pGeom As IGeometry

    Set pGeom = pArea

    Dim pPCS As IProjectedCoordinateSystem

    Set pPCS = pGeom.SpatialReference

    Dim dFactor As Double

    dFactor = pPCS.CoordinateUnit.MetersPerUnit * pPCS.CoordinateUnit.MetersPerUnit

    Dim dAcresPerSqMeter As Double

    dAcresPerSqMeter = 0.0002471

    GetArea = dAcresPerSqMeter * dFactor * pArea.Area

End Function

Function ClipPoly(pTopoOp As ITopologicalOperator, pOtherGeom As IPolygon) As IPolygon

    Dim pPolygon As IPolygon

    Simplify pOtherGeom

    Dim pGeom As IGeometry

    Set pGeom = pTopoOp

    Set pPolygon = pTopoOp.Intersect(pOtherGeom, esriGeometry2Dimension)

    Simplify pPolygon

    Set ClipPoly = pPolygon

End Function

Sub Simplify(pTopoOp2 As ITopologicalOperator2)

    pTopoOp2.IsKnownSimple = False

    pTopoOp2.Simplify

End Sub

